# Maintainer: Balló György <ballogyor+arch at gmail dot com>
# Contributor: Thorsten Töpper <atsutane-tu@freethoughts.de>

pkgname=glyr
pkgver=1.0.10
pkgrel=5
pkgdesc='Music metadata searchengine utility and library written in C'
arch=('loong64' 'x86_64')
url='https://github.com/sahib/glyr'
license=('LGPL3')
depends=('curl' 'glib2' 'sqlite')
makedepends=('cmake' 'ninja')
source=("https://github.com/sahib/$pkgname/archive/$pkgver/$pkgname-$pkgver.tar.gz")
sha256sums=('77e8da60221c8d27612e4a36482069f26f8ed74a1b2768ebc373c8144ca806e8')

build() {
  cmake -S $pkgname-$pkgver -B build -G Ninja -DCMAKE_INSTALL_PREFIX='/usr'
  cmake --build build
}

package() {
  DESTDIR="$pkgdir" cmake --install build
}
